Table of Contents

Why are Lookup Formulas so Common in Spreadsheets?

1) Questions:
   - Why are Lookup Formulas so common in Spreadsheets?
2) Answer:
   - Because everybody in all industries and endeavors has to look things up in tables!

Examples of Common Lookup Situations:

3) For an invoice, you have to look up the price for the Quad boomerang:

Boomerang	Part Number	Flight Range (meters)	Price
Bellen	1000-165-B100	25	$26.95
Carlota	1001-540-C101	20	$28.95
Majestic Beaut	1002-394-M102	35	$31.95
Quad	1003-307-Q103	20	$35.95
Sunshine	1004-848-S104	30	$18.95
Sunset	1005-155-S105	40	$20.95
Tri-Fly	1006-552-T106	1	$4.95
Outdoor Tri-Fly	1007-634-O107	5	$8.95

4) If Jo had sales of $8,000 what is her commission pay?

Sales Amount	Category	Commission Paid
$0.00	Sub Par	$0.00
$1,000.00	Par	$20.00
$2,500.00	Above Par	$100.00
$7,000.00	Very Good	$250.00
$10,000.00	Excellent	$700.00

5) If taxable earnings are $2500, what is tax paid?

Over --	But not over --	Rate	Tax from Previous brackets	Rule
$0	$1,313	0%	$0.00	Zero Tax
$1,313	$2,038	10%	$0.00	10% of excess over $1,313
$2,038	$6,304	15%	$72.50	$72.50 + 15% of excess over $2,038
$6,304	$9,844	25%	$712.40	$712.40 + 25% of excess over $6,304
$9,844	$18,050	28%	$1,597.40	$1,597.40 + 28% of excess over $9,844
$18,050	$31,725	33%	$3,895.08	$3,895.08 + 33% of excess over $18,050
$31,725		35%	$8,407.83	$8,407.83 + 35% of excess over $31,725

6) What is Pearlie's phone?

ID	Last	First	E-mail	Phone
880-10000	Leff	Julianne	LeffJ@PBY.com	253-559-4034
880-10001	Piano	Milagros	PianoM@PBY.com	253-553-4381
880-10002	Coller	Kathrine	CollerK@PBY.com	206-762-2195
880-10003	Stackpole	Lonnie	StackpoleL@PBY.com	253-764-6538
880-10004	Lintz	Kurt	LintzK@PBY.com	206-736-4510
880-10005	Dudgeon	Penelope	DudgeonP@PBY.com	253-452-9723
880-10006	Hughs	Pearlie	HughsP@PBY.com	253-719-7600

7) What size drainage pipe do we use for 6000 square foot lot?

Square footage of Lot	Pipe diameter
0	6"
500	8"
1000	10"
5000	16"
10000	25"
25000	35"
50000	40"

Lookup Tables are Usually Vertical (Why VLOOKUP Function is so Common)

1) If you are trying to find the price for "Quad" boomerang in the lookup table below we have to look through a ***vertical*** list of boomerang names in the first column of the lookup table.
    - The table is ***Vertical*** because the boomerang names are listed vertically, one on top of the other, each in a different row.
    - Most lookup tables that people use are oriented in this ***vertical*** fashion.

Boomerang	Part Number	Flight Range (meters)	Price
Bellen	1000-165-B100	25	$26.95
Carlota	1001-540-C101	20	$28.95
Majestic Beaut	1002-394-M102	35	$31.95
Quad	1003-307-Q103	20	$35.95
Sunshine	1004-848-S104	30	$18.95
Sunset	1005-155-S105	40	$20.95
Tri-Fly	1006-552-T106	1	$4.95
Outdoor Tri-Fly	1007-634-O107	5	$8.95

2) Occasionally you will see a horizontally oriented lookup table, like when a manager who is not familiar with the concept of a Proper Data Set. If we needed to look up Tyrone's phone, we would have to do a "Horizontal lookup":

Category	Tina	Gigi	Chin	Tyrone	Bobi
Hire Date	1/4/2005	3/28/2009	6/11/2012	5/24/2010	2/2/2014
Salary	$93,976.00	$40,233.00	$36,762.00	$89,589.00	$52,319.00
Office Number	302	348	383	230	255
Phone	(206) 311-2567	(253) 577-9177	(206) 582-3391	(253) 561-5768	(253) 598-1171

Although Most Lookup Tables are Vertical, it is Not Uncommon to have Unusual Lookup Situations

1) Example, we need to look up vendor's name for the lowest bid:

	A	B	C	D	E	F	G	H	I	J
150	**Example 10:** INDEX and MATCH to find Vendor for Low Bid.									
151										
152	**Venders and bids:**									
153	Crank'ys	Bay Air	Compressor R	Mech-Aid	Han's Shop	Low Bid	Vendor for Low Bid			
154	$38.99	$48.60	$43.53	$40.08	$47.92	$38.99	Crank'ys			
155	$57.68	$31.80	$52.78	$31.42	$55.19	$31.42	Mech-Aid			
156	$53.32	$32.64	$37.69	$48.29	$41.59	$32.64	Bay Air			
157	$35.20	$40.55	$32.65	$36.81	$41.14	$32.65	Compressor R Us			
158	$56.72	$47.16	$36.42	$49.56	$39.25	$36.42	Compressor R Us			
159	$47.91	$35.08	$51.13	$49.84	$42.12	$35.08	=INDEX($A$153:$E$153,MATCH(F159,A159:E159,0))			
160	$34.81	$35.11	$48.63	$33.32	$37.83	$33.32	Mech-Aid			
161	$42.25	$35.76	$58.60	$46.28	$40.53	$35.76	Bay Air			
162	$40.14	$42.31	$37.62	$59.97	$42.57	$37.62	Compressor R Us			
163	$36.48	$40.79	$53.24	$51.01	$51.24	$36.48	Crank'ys			
164	$38.57	$40.06	$54.71	$39.70	$54.73	$38.57	Crank'ys			
165	$52.66	$43.61	$59.98	$34.61	$52.65	$34.61	Mech-Aid			

Lookup Functions

2) VLOOKUP
    - Does vertical lookup, trying to match a value in the first column of a lookup table and then retrieve an item from a subsequent column. Does Exact or Approximate Match. Since most lookup tables are vertical, VLOOKUP is the most commonly used lookup function.
3) HLOOKUP
    - Does horizontal lookup, Exact or Approximate Match. This function is not commonly used.
4) LOOKUP
    - Does vertical or horizontal lookup. Only does Approximate Match. Since this function can handle array operations and VLOOKUP cannot, it is often used for Lookup Array Formulas.
5) MATCH
    - Returns the relative position of an item in a horizontal or vertical list (one-way array, either vertical or horizontal)
6) INDEX
    - a lookup function that can do a two-way lookup or one-way lookup and even lookup a whole row or column. INDEX and MATCH together are the most versatile lookup functions.
7) CHOOSE
    - Can choose items from a list based on an index number. This function is useful for situations where you have multiple lookup tables.

VLOOKUP Function:

1) V means Vertical
2) The goal of VLOOKUP is:
    - To go and get a value and deliver it to a cell or formula.
    - Try to match a value in the first column of a lookup table and then retrieve an item from a subsequent column and returns that item to the cell or formula.
3) VLOOKUP arguments: VLOOKUP (**lookup_value** , **table_array** , **col_index_num** , **[range_lookup]**)
    - **lookup_value**:
        1. The value you want to look up and try to match against the items in the first column of the lookup table.
    - **table_array**:
        1. This is the lookup table.
        2. The first column in the cell range must contain the lookup_value
    - **col_index_num**:
        1. Column number in the lookup table that holds the value you want to return to the cell.
        2. If price is in the 4th column of the lookup table, use 4 to indicate that the 4th column holds the price.
    - **[range_lookup]**:
        1. This argument tells VLOOKUP what type of lookup you are doing: Exact or Approximate.
        2. Exact Match:
            i. Use FALSE or 0.
            ii. VLOOKUP will do a linear search from the first item in the first column of the lookup table and keep searching until it finds an exact match.
            iii. If there are duplicates, it only finds the first one.
            iv. If it can't find a match, it returns an #N/A error.
        3. Approximate Match
            i. Use TRUE or 1 or omitted.
            ii. First column must be sorted ascending (biggest to smallest).
                1. Sorted either numerically or alphabetically
            iii. Approximate Match is what you use when you are looking up:
                1. Tax Rates
                2. Commission Rates
                3. Decimal Grades
                4. Drainage Pipe Size
            iv. Metaphor for understanding how Approximate Match works:
                1. It starts its search at the first item in first column, and looks at each one and when it bumps into first bigger value, it jumps back one row.
            v. Technically, Approximate Match finds the largest value in the sorted first column that is less than or equal to lookup_value by using a Binary search that works this way:
                1. Binary Search reduces search time because it repeatedly divides the table in half and checks the one in the middle.
                2. It doesn't have to check each one.
                3. Binary Search and is faster than Linear Search (Exact Match)
            vi. If lookup_value is less than 1st value in table VLOOKUP returns #N/A

VLOOKUP with Exact Match and IF and ISBLANK to Create Invoice Example

1) VLOOKUP with Exact Match for looking up the price of a boomerang product.
2) Because VLOOKUP can look at empty cells in our invoice and we don't want #N/A, we can use 2 options:
    1. =IFERROR(VLOOKUP(A16,$A$4:$D$11,4,0),"") or
       =IFNA(VLOOKUP(A16,$A$4:$D$11,4,0),"")
       Advantage: it is an easier formula to create than using IF and ISBLANK.
    2. =IF(ISBLANK(A16),"",VLOOKUP(A16,$A$4:$D$11,4,0))
        i. Advantage: for spreadsheets with many VLOOKUP formulas that are doing Exact Match and that have very large lookup tables, formula calculation would slow down if we used the IFERROR formulas. With the IF and ISBLANK, when the cell is empty, the VLOOKUP part of the formula never has to run because the IF just puts a "zero-length text string ("") into the cell.

	A	B	C	D	E
1	**Example 1:** Data Validation Drop-down list to Select Boomerang Name. VLOOKUP with Exact Match to lookup price for Invoice:				
2					
3	Boomerang	Part Number	Flight Range (meters)	Price	
4	Bellen	1000-165-B100	25	$26.95	
5	Carlota	1001-540-C101	20	$28.95	
6	Majestic Beaut	1002-394-M102	35	$31.95	
7	Quad	1003-307-Q103	20	$35.95	
8	Sunshine	1004-848-S104	30	$18.95	
9	Sunset	1005-155-S105	40	$20.95	
10	Tri-Fly	1006-552-T106	1	$4.95	
11	Outdoor Tri-Fly	1007-634-O107	5	$8.95	
12					
13					
14	**Invoive Example:**				
15	Product	Units	Price	Total	
16	Carlota	5	=IF(ISBLANK(A16),"",VLOOKUP(A16,$A$4:$D$11,4,0))		
17	Sunshine	2	18.95	37.9	
18					
19					
20					
21					
22			**Subtotal**	182.65	

VLOOKUP with Approximate Match to Lookup Pipe size:

	A	B	C
39	Square footage of Lot	Pipe diameter	
40	0	6"	
41	500	8"	
42	1,000	10"	
43	5,000	16"	
44	10,000	25"	
45	25,000	35"	
46	50,000	40"	
47			
48	Square footage of Lot	15,888	
49	Pipe diameter (inches)	=VLOOKUP(B48,A40:B46,2)	
50		VLOOKUP(lookup_value, table_array, **col_index_num**, [range_lookup])	

VLOOKUP with Cell Reference for Column Index Number



	A	B	C	D	E
66					
67			2	3	
68	Employee	Sales	Rating	Commission Paid	
69	Sioux	$7,598.00	Very Good	$250.00	
70	Kim	$68.00	Sub Par	$0.00	
71	Gigi	$15,980.00	Excellent	$700.00	
72	Franny	$2,499.99	Par	=VLOOKUP($B72,$A$55:$C$59,D$67)	
73					

VLOOKUP(lookup_value, table_array, **col_index_num**, [range_lookup])

VLOOKUP with MATCH function for Column Index Number.



	A	B	C	D	E
77	ID	Last	First	E-mail	Phone
78	880-10000	Leff	Julianne	LeffJ@PBY.com	253-559-4034
79	880-10001	Piano	Milagros	PianoM@PBY.com	253-553-4381
80	880-10002	Coller	Kathrine	CollerK@PBY.com	206-762-2195
81	880-10003	Stackpole	Lonnie	StackpoleL@PBY.com	253-764-6538
82	880-10004	Lintz	Kurt	LintzK@PBY.com	206-736-4510
83	880-10005	Dudgeon	Penelope	DudgeonP@PBY.com	253-452-9723
84	880-10006	Hughs	Pearlie	HughsP@PBY.com	253-719-7600
85					
86	MATCH function is a lookup function.				
87	MATCH function returns the relative position of an item in a list.				
88	This method is dynamic: 1) columns can be in any order and 2) If you change input formula updates.				
89					
90					
91	ID	First	Last	E-mail	
92	880-10002	=VLOOKUP($A92,$A$78:$E$84,MATCH(B91,$A$77:$E$77,0),0)			
93					

VLOOKUP(lookup_value, table_array, **col_index_num**, [range_lookup])

VLOOKUP with ROWS function for Column Index Number to Retrieve Record and List Vertically.



	A	B
97	ID	
98	880-10002	
99	=VLOOKUP($A$98,$A$78:$E$84,ROWS(A$98:A99),0)	
100	Kathrine	
101	CollerK@PBY.com	
102	206-762-2195	
103		

VLOOKUP(lookup_value, table_array, **col_index_num**, [range_lookup])

VLOOKUP with COLUMNS function for Column Index Number to Retrieve Record and List Horizontally.



	C	D	E	F	G	H	I	J	K	L
97	ID									
98	880-10002	Coller	Kathrine	CollerK@PBY.com	=VLOOKUP($A$98,$A$78:$E$84,COLUMNS($C$98:G98),0)					
99										

Multiple VLOOKUP Functions to Calculate Taxes:

	A	B	C	D	E	F
104	**Tax Example:**					
105		**Married**				
106	**Lookup Value**	**Over --**	**But not over --**	**Rate**	**Tax from Previous brackets**	**Rule**
107	$0	$0	$1,313	0%	$0.00	Zero Tax
108	$1,313.01	$1,313	$2,038	10%	$0.00	10% of excess over $1,313
109	$2,038.01	$2,038	$6,304	15%	$72.50	$72.50 + 15% of excess over $2,038
110	$6,304.01	$6,304	$9,844	25%	$712.40	$712.40 + 25% of excess over $6,304
111	$9,844.01	$9,844	$18,050	28%	$1,597.40	$1,597.40 + 28% of excess over $9,844
112	$18,050.01	$18,050	$31,725	33%	$3,895.08	$3,895.08 + 33% of excess over $18,050
113	$31,725.01	$31,725		35%	$8,407.83	$8,407.83 + 35% of excess over $31,725
114						
115	Taxable Earnings	$15,896.00				
116	Tax from Previous brackets	$1,597.40	=VLOOKUP($B$115,$A$107:$E$113,5)			
117	Tax Rate	28.00%	=VLOOKUP($B$115,$A$107:$E$113,4)			
118	Amount already taxed	$9,844.00	=VLOOKUP($B$115,$A$107:$E$113,2)			
119	Amount to tax in this bracket	$6,052.00	=B115-B118			
120	Total Tax	$3,291.96	=ROUND(B119*B117,2)+B116			
121						
122	Taxable Earnings w LOOKUP					
123	$3,291.96	=VLOOKUP($B$115,$A$107:$E$113,5)+ROUND((B115-VLOOKUP($B$115,$A$107:$E$113,2))*VLOOKUP($B$115,$A$107:$E$113,4),2)				
124	$3,291.96	=LOOKUP(B115,A107:E113)+ROUND((B115-LOOKUP(B115,A107:B113))*LOOKUP(B115,A107:D113),2)				
125						

VLOOKUP, LEFT and SEARCH to do a "Partial Text Lookup. Goal: Lookup Product Price.

	A	B	C	D	E	F
128	ID	Price			ID	Price
129	Bellen-234-B25R	=VLOOKUP(LEFT(A129,SEARCH("-",A129)-1),$E$129:$F$131,2,0)			Bellen	$26.00
130	Carlota-345-C20R				Carlota	$23.00
131	Quad-765-Q20L	$36.00			Quad	$36.00
132						

VLOOKUP(**lookup_value**, table_array, col_index_num, [range_lookup])

HLOOKUP Function:

1) H means Horizontal
2) Same as VLOOKUP, but it finds a match in the first row and returns an item from a specified row number.
3) Example:

	A	B	C	D	E	F
27	Category	Tina	Gigi	Chin	Tyrone	Bobi
28	Hire Date	1/4/2005	3/28/2009	6/11/2012	5/24/2010	2/2/2014
29	Salary	$93,976.00	$40,233.00	$36,762.00	$89,589.00	$52,319.00
30	Office Number	302	348	383	230	255
31	Phone	(206) 311-2567	(253) 577-9177	(206) 582-3391	(253) 561-5768	(253) 598-1171
32						
33	Employee:	Gigi				
34	Phone:	=HLOOKUP(B33,B27:F31,5,0)				
35						

HLOOKUP(**lookup_value**, table_array, row_index_num, [range_lookup])

MATCH function

1) What it does:
    - Searches for an item in a list and returns the relative position of the item in the list.
    - More specifically: Returns the relative position of an item in a horizontal or vertical list (one-way array, either vertical or horizontal)
2) MATCH(lookup_value, lookup_array, [match_type])
    - **lookup_value**:
        1. The value you want to look up.
        2. The value you want to look up must be in the range of cells you specify in lookup-array.
    - **lookup_array**:
        1. The range of cells being searched.
        2. Must be a one-way array, either:
            i. Vertical (one column with one or more rows)
            ii. Horizontal (one row with one or more columns)
    - **[match_type]**:
        1. Optional. The numbers: -1, 0, or 1.
        2. The match_type argument specifies how Excel matches lookup_value with values in lookup_array.
        3. The default value for this argument is 1
        4. **1 or omitted**
            i. MATCH finds the largest value that is less than or equal to lookup_value. The values in the lookup_array argument must be placed in ascending order.
            ii. Works just like VLOOKUP Approximate Match
        5. **0**
            i. MATCH finds the first value that is exactly equal to lookup_value. The values in the lookup_array argument can be in any order.
            ii. Works just like VLOOKUP Exact Match
        6. **-1**
            i. MATCH finds the smallest value that is greater than or equal to lookup_value. The values in the lookup_array argument must be placed in descending order.
        7. Another alternative way to describe this argument:
            i. **[match_type]** tells the MATCH what sort of lookup to do:
            ii. 1 or empty = approximate match; table sorted ascending; first bigger value bumped into then jump back one position, if value is smaller than first item returns #N/A, if bigger than last it returns last value
            iii. 2 = extract match, if duplicates, it finds first one only, can't find it shows #N/A
            iv. -1 = approximate match; table sorted descending; first smaller value bumped into then jump back one position, if value is bigger than first item returns #N/A, if smaller than last it returns last value

MATCH to Compare Two Lists

	A	B	C	D	E	F	G	H	I	J
1	Has prospective customer made it into our master list due to our sales phone calls?									
2										
3			**List 2 = Customers we have made sales calls to.**							
4	Master Customer List		Prospective Customers							
5	List 1		List 2	Is item in List 2 in List 1?				Is item in List 2 NOT in List 1?		
6	Timothy Ramsey		Maria Carpenter	TRUE		=ISNUMBER(MATCH(C6,$A$6:$A$15,0))		FALSE		=ISNA(MATCH(C6,$A$6:$A$15,0))
7	Maria Carpenter		Jeff Lyons	TRUE		=ISNUMBER(MATCH(C7,$A$6:$A$15,0))		FALSE		=ISNA(MATCH(C7,$A$6:$A$15,0))
8	Jeff Lyons		Judy Allen	TRUE		=ISNUMBER(MATCH(C8,$A$6:$A$15,0))		FALSE		=ISNA(MATCH(C8,$A$6:$A$15,0))
9	Judy Allen		Fred Mixxer	FALSE		=ISNUMBER(MATCH(C9,$A$6:$A$15,0))		TRUE		=ISNA(MATCH(C9,$A$6:$A$15,0))
10	Daniel Campbell		Eugene Conner	TRUE		=ISNUMBER(MATCH(C10,$A$6:$A$15,0))		FALSE		=ISNA(MATCH(C10,$A$6:$A$15,0))
11	Eugene Conner		Peter Owen	TRUE		=ISNUMBER(MATCH(C11,$A$6:$A$15,0))		FALSE		=ISNA(MATCH(C11,$A$6:$A$15,0))
12	Peter Owen		Jason Keller	TRUE		=ISNUMBER(MATCH(C12,$A$6:$A$15,0))		FALSE		=ISNA(MATCH(C12,$A$6:$A$15,0))
13	Rhonda Walsh		Min Pham	FALSE		=ISNUMBER(MATCH(C13,$A$6:$A$15,0))		TRUE		=ISNA(MATCH(C13,$A$6:$A$15,0))
14	Jason Keller									

INDEX function

1) Two possible uses for INDEX function:

```
INDEX(array, row_num, [column_num])
INDEX(reference, row_num, [column_num], [area_num])
```

2) We will ONLY use the first one with the arguments: **array**, **row-num** and **[column_num]**.
3) What it does:
    - INDEX and MATCH together are the most versatile lookup functions because they can do anything that VLOOKUP can do, plus much more such as:
        1. It can do two-way lookup: where we are retrieving an item from a table at the intersection of a row header and column header.
        2. It can do one-way lookup on a column of values (column filled with rows)
        3. It can do one-way lookup on a row of values (row filled with columns)
        4. It can look up a whole column of values
        5. It can look up a whole row of values
4) Arguments:
    - **array**:
        1. A two dimensional table (both a row and a column).
           or
        2. A one dimensional table (row or column).The range/array containing the values you want to look up.
    - **row_num**
        1. Tells INDEX from which row to retrieve the item.
        2. If you put a 0 (zero) or omitted, all the rows are returned; this is how you “lookup a whole column”.
        3. If the array argument holds a one-way array (column or row), you can put the relative position into the **row_num** argument.
    - **column_num**
        1. Tells INDEX from which column to retrieve the item.
        2. If you put a 0 (zero) or omitted, all the columns are returned; this is how you “lookup a whole row”.
5) When INDEX looks up a row or column, it returns a range of values, not an array. This is why we can add with SUM.

INDEX and MATCH functions to “Lookup Left” (Lookup in First Column of Lookup Table)



	A	B	C	D
140	Use INDEX as a substitute for VLOOKUP when the value you want to return is to the left of the lookup column.			
141	**Product**	**Price**	**Description**	
142	V Range	$12.95	Flying Range is 10	
143	Quad	$39.95	Flying Range is 20	
144	Weighted Sunset	$40.00	Flying Range is 50	
145	Weighted MB	$45.00	Flying Range is 60	
146	S Rang	$65.00	Flying Range is 70	
147	Long D	$69.00	Flying Range is 80	
148	Long D Squared	$100.00	Flying Range is 85	
149	Dan's Special	$110.00	Flying Range is 110	
150	Manu Magic	$165.00	Flying Range is 160	
151				
152		1-Way Lookup Vertical: Lookup Column to Left		
153		Description	Product	
154		Flying Range is 110	=INDEX(A142:A150,MATCH(B154,C142:C150,0))	
155				
156				

INDEX and MATCH functions to lookup Vendor Low Bid

	A	B	C	D	E	F	G	H	I	J
159	**Venders and bids:**									
160	Crank'ys	Bay Air	Compressor R U	Mech-Aid	Han's Shop	Low Bid	Vendor for Low Bid			
161	$38.99	$48.60	$43.53	$40.08	$47.92	$38.99	Crank'ys			
162	$57.68	$31.80	$52.78	$31.42	$55.19	$31.42	Mech-Aid			
163	$53.32	$32.64	$37.69	$48.29	$41.59	$32.64	Bay Air			
164	$35.20	$40.55	$32.65	$36.81	$41.14	$32.65	=INDEX($A$160:$E$160,MATCH(F164,A164:E164,0))			
165	$56.72	$47.16	$36.42	$49.56	$39.25	$36.42	C			
166	$47.91	$35.08	$51.13	$49.84	$42.12	$35.08	Bay Air			
167	$34.81	$35.11	$48.63	$33.32	$37.83	$33.32	Mech-Aid			
168	$42.25	$35.76	$58.60	$46.28	$40.53	$35.76	Bay Air			
169	$40.14	$42.31	$37.62	$59.97	$42.57	$37.62	Compressor R Us			
170	$36.48	$40.79	$53.24	$51.01	$51.24	$36.48	Crank'ys			
171	$38.57	$40.06	$54.71	$39.70	$54.73	$38.57	Crank'ys			
172	$52.66	$43.61	$59.98	$34.61	$52.65	$34.61	Mech-Aid			

INDEX(array, row_num, [column_num])
INDEX(reference, row_num, [column_num], [area_num])

INDEX and MATCH functions to do a Two-Way Lookup

	A	B	C	D	E	F
177	2-Way lookup					
178	Discount Table					
179	PartID/Qty	1	5	25	100	
180	**Part1**	13.00%	15.00%	17.00%	19.00%	
181	**Part2**	14.00%	16.00%	18.00%	20.00%	
182	**Part3**	15.00%	17.00%	19.00%	21.00%	
183	**Part4**	16.00%	18.00%	20.00%	22.00%	
184	**Part5**	17.00%	19.00%	21.00%	23.00%	
185	**Part6**	18.00%	20.00%	22.00%	24.00%	
186						
187	PartID	**Part5**	5	<<== Row	=MATCH(B187,A180:A185,0)	
188	Qty	15	2	<<==Column	=MATCH(B188,B179:E179)	
189	Discount	19.00%		=INDEX(B180:E185,C187,C188)		
190	Discount	19.00%		=INDEX(B180:E185,MATCH(B187,A180:A185,0),MATCH(B188,B179:E179))		

INDEX and MATCH to lookup Whole Column of Values.

	A	B	C	D	E
195	Lookup whole row or column				
196	Use INDEX, MATCH and SUM to retrieve a whole column of values and add them				
197	0 (zero) or empty in row_num argument will return a whole column of values (“all rows” or “entire column”)				
198	Month to add	Mar Total			
199	Mar	=SUM(INDEX(A201:D203,,MATCH(A199,A200:D200,0)))			
200	Jan	Feb	Mar	Apr	
201	1	4	4	5	
202	3	4	12	1	
203	1	4	2	2	

INDEX(array, **row_num**, [column_num])
INDEX(reference, **row_num**, [column_num], [area_num])

CHOOSE function

1) CHOOSE is a lookup function that can return “things” to a cell or a formula
2) "Things" you are allowed to look up and return to a cell or formula:
    - Text
    - Numbers
    - Formulas
    - Functions
    - Cell References
    - Ranges
    - Defined Names
    - Array Constants
3) Function: =CHOOSE(**index_num**, **value1**,**value2**, **…**)
4) Arguments:
    - index_num
        1. Must be an index number like: 1,2,3,4,5,…
    - value1, value2,…
        1. These are the “things" you want to return to cell or formula
        2. You actually store the “things” that want to look up in the value1, value2 arguments.
            i. Unlike other lookup functions, CHOOSE requires that you enter the "things" into the function individually

CHOOSE function To Select the Correct Table for VLOOKUP

- CHOOSE functions allows you to lookup things like different lookup tables.
- =CHOOSE(**index_num**, **value1**,**value2**, …)
    1. index_num is a number like 1, 2, 3, 4
    2. value1 = lookup table 1
    3. value2 = lookup table 2
    4. value3 = lookup table 3
    5. If index_num is 2, then the lookup table 2 will be returned by CHOOSE to the VLOOKUP function

	A	B	C	D	E	F	G
226	**VLOOKUP & CHOOSE with 3 lookup tables**					ABC	
227						Units Sold	Commission Rate
228	Product	Units Sold	Table	Commission Rate		0	1.00%
229	ABC	431	1	4.00%		100	2.00%
230	EDR	65	2	=VLOOKUP(B230,CHOOSE(C230,$F$228:$G$231,$F$235:$G$238,$F$242:$G$245),2)			
231	EDS	563	3	VLOOKUP(lookup_value, **table_array**, col_index_num, [range_lookup])		500	6.00%
232	ABC	493	1	4.00%			
233	EDS	188	3	2.00%		EDR	
234						Units Sold	Commission Rate
235						0	1.00%
236	ABC	1		Commission Rate		200	2.00%
237	EDR	2		4.00%		300	4.00%
238	EDS	3		1.00%		400	6.00%
239				4.00%			
240				4.00%		EDS	
241				2.00%		Units Sold	Commission Rate
242						0	2.00%
243	**Reference video:**					300	3.00%
244	CHOOSE Function Beginner to Advanced 12 Examples (Excel VLOOKUP WEEK Video #5)					500	4.00%
245	http://www.youtube.com/watch?v=M4X2SXdXWmE					750	5.00%

LOOKUP Function:

1) LOOKUP is THE original Lookup function from the first spreadsheet invented (VisiCalc). This function was invented before VLOOKUP.
2) Two possible uses for LOOKUP function. Screen Tip shows the arguments for the two possibilities:



3) What LOOKUP function does:
    - Goes and gets an item and brings it back to the cell or formula.
    - LOOKUP can do either Vertical or Horizontal lookup.
    - It can ONLY do Approximate Match.
4) ***Approximate Match ONLY***
    - If you sort (smallest to biggest) the first column of the array or the lookup_vector, you can force LOOKUP into doing Exact Match.
5) If you use "**lookup_value**" and "**array**" arguments:
    - **array** is the lookup table, where the first column must be sorted (smallest to biggest)
    - When you use **array** as your lookup table, LOOKUP does either Vertical or Horizontal lookup
    - If the number of rows in the lookup table are greater than or equal to the number of columns, LOOKUP does vertical lookup
        1. When doing vertical lookup, it always uses the last column as the column that contains the values it will potentially return to the cell or formula (this is the reason we do not need to put a column index number into LOOKUP).
    - If the number of columns in the lookup table are greater than the number of rows, LOOKUP does horizontal lookup
        1. When doing horizontal lookup, it always uses the last row as the row that contains the values it will potentially return to the cell or formula (this is the reason we do not need to put a row index number into LOOKUP).
6) If you use "**lookup_value**" and "**lookup_vector**" and "**result_vector**" arguments:
    - The **lookup_vector** argument is like the MATCH function because it will deliver a relative position to the LOOKUP function that will be used to pick put an item from the **result_vector** argument. Said a different way: LOOKUP will find the position of the "lookup_value" in the "lookup_vector" to find the relative position, and that retrieve an item from the "result_vector" in that relative position.
    - These arguments can hold array operations without using Ctrl + Shift + Enter:
        1. The **lookup_vector** argument in the LOOKUP function
        2. The **result_vector** argument in the LOOKUP function
7) There are two good reasons we might want to use LOOKUP rather than VLOOKUP:
    - We are doing Approximate Match lookup and we want to enter fewer arguments than if we were to use VLOOKUP (LOOKUP doesn't need column index number like VLOOKUP)
    - We are doing Approximate Match lookup and need to make an array calculation, which VLOOKUP can NOT do.

LOOKUP Doesn't Need Column Index Number Like VLOOKUP

	A	B	C	D	E
260	Sales Amount	Rating	Commission Paid		
261	$0.00	Sub Par	$0.00		
262	$1,000.00	Par	$20.00		
263	$2,500.00	Above Par	$100.00		
264	$7,000.00	Very Good	$250.00		
265	$10,000.00	Excellent	$700.00		
266					
267	**Example 15:** LOOKUP with **lookup_value** and **array** arguments. Approximate Match. **Goal:** Lookup Commission and Rating.				
268	**Reason we use LOOKUP rather than VLOOKUP: Fewer Arguments to Enter.**				
269	**If Jo had sales of $8,000 what is her commission pay?**				
270	Name	Sales	Commission Paid	Rating	
271	Jo	$8,000.00	250	=LOOKUP(B271,A261:B265)	
272					

LOOKUP(lookup_value, **lookup_vector**, [result_vector])
LOOKUP(lookup_value, **array**)

LOOKUP Can Make Array Calculations with no Ctrl Shift + Enter. LOOKUP Can Have a lookup_vector and a result_vector.

	A	B	C
277	Get Last Site based on Last Date	=LOOKUP(2,1/ISNUMBER(A280:A290),B280:B290)	
278			
279	Date	Site Visited	
280	5/1/2016	3455 240th St., Des Monies	
281	5/1/2016	1254 10th St., Seattle	
282		1254 10th St., Seattle	
283	5/2/2016	478 Birch St., Tukwila	
284	5/5/2016		
285	5/5/2016	3455 240th St., Des Monies	
286		1254 10th St., Seattle	
287			
288			
289			
290			
291			

LOOKUP(lookup_value, lookup_vector, **[result_vector]**)
LOOKUP(lookup_value, array)

SUMPRODUCT and LOOKUP Can Do "Lookup Adding"

	A	B	C	D	E	F	G	H	I	J	K
9	**Data set with number of rolls of fence sold for each transaction:**								**Pricing Table:**		
10											
11	Date	Number of Rolls	Price	Revenue		Total Revenue Helper Column	Total Revenue Single Cell Formula		Units From	Units To	Price per Roll (100 feet)
12	9/16/2015	540	$168.00	$90,720.00		$11,852,436.00	=SUMPRODUCT(LOOKUP(B12:B183,I12:K15),B12:B183)				$198.00
13	9/15/2015	360	$168.00	$60,480.00					144	288	$187.00
14	9/16/2015	12	$198.00	$2,376.00					289	577	$168.00
15	9/15/2015	468	$168.00	$78,624.00					578	and up	$152.00
16	9/17/2015	48	$198.00	$9,504.00							
17	9/18/2015	84	$198.00	$16,632.00							
18	9/17/2015	540	$168.00	$90,720.00							
19	9/18/2015	492	$168.00	$82,656.00							
20	9/18/2015	204	$187.00	$38,148.00							

SUMPRODUCT(array1, [array2], [array3], [array4], ...)

Difference between Exact Match and Approximate Match for Lookup functions:

1) Exact Match:
    - Does a linear search from the first item and keep searching until it finds an exact match.
    - If there are duplicates, it only finds the first one.
    - If it can't find a match, it returns an #N/A error.
2) Approximate Match
    - Range must be sorted ascending (biggest to smallest).
        1. Sorted either numerically or alphabetically
    - Approximate Match is what you use when you are looking up:
        1. Tax Rates
        2. Commission Rates
        3. Decimal Grades
        4. Drainage Pipe Size
    - Approximate match finds the largest value that is less than or equal to lookup_value.
    - Metaphor for understanding how it works:
        1. It starts its search at the first item, and looks at each item, one at a time and when it bumps into first bigger value, it jumps back one position.
    - Binary search is how it really works.
        1. Binary Search and is faster than Linear Search (Exact Match)
        2. Binary Search reduces search time because it repeatedly divides the table or range in half and checks the one in the middle to help reduce calculation time.
        3. It doesn't have to check each one.
    - If lookup_value is less than 1st value in table or range it will return an #N/A

Use wildcard characters

3) If you are doing an Exact Match and the lookup_value is text, you can use the wildcard characters in lookup_value.
    - Wildcard Characters:
        1. Question mark (?)
            i. A question mark matches any single character.
        2. Asterisk (*)
            i. An asterisk matches any sequence of characters (zero or more characters).
    - If you want to find an actual question mark or asterisk, type a tilde (~) in front of the character.

Cumulative List of Keyboards Throughout Class:

1) Esc Key:
    i. Closes Backstage View (like Print Preview).
    ii. Closes most dialog boxes.
    iii. If you are in Edit mode in a Cell, Esc will revert back to what you had in the cell before you put the Cell in Edit mode.
2) **F2 Key** = Puts formula in Edit Mode and shows the rainbow colored Range Finder.
3) SUM Function: **Alt + =**
4) **Ctrl + Shift + Arrow** = Highlight column (Current Region).
5) **Ctrl + Backspace** = Jumps back to Active Cell
6) **Ctrl + Z** = Undo.
7) **Ctrl + Y** = Undo the Undo.
8) **Ctrl + C** = Copy.
9) **Ctrl + X** = Cut.
10) **Ctrl + V** = Paste.
11) **Ctrl + PageDown** =expose next sheet to right.
12) **Ctrl + PageUp** =expose next sheet to left.
13) **Ctrl + 1** = Format Cells dialog box, or in a chart it opens Format Chart Element Task Pane.
14) **Ctrl + Arrow**: jumps to the bottom of the "**Current Region**", which means it jumps to the last cell that has data, right before the first empty cell.
15) **Ctrl + Home** = Go to Cell A1.
16) **Ctrl + End** = Go to last cell used.
17) Alt keyboards are keys that you hit in succession. Alt keyboards are keyboards you can teach yourself by hitting the Alt key and looking at the screen tips.
    i. Create PivotTable dialog box: **Alt, N, V**
    ii. Page Setup dialog box: **Alt, P, S, P**
    iii. Keyboard to open Sort dialog box: **Alt, D, S**
18) **ENTER** = When you are in Edit Mode in a Cell, it will put thing in cell and move selected cell DOWN.
19) **CTRL + ENTER** = When you are in Edit Mode in a Cell, it will put thing in cell and keep cell selected.
20) **TAB** = When you are in Edit Mode in a Cell, it will put thing in cell and move selected cell RIGHT.
21) **SHIFT + ENTER** = When you are in Edit Mode in a Cell, it will put thing in cell and move selected cell UP.
22) **SHIFT + TAB** = When you are in Edit Mode in a Cell, it will put thing in cell and move selected cell LEFT.
23) **Ctrl + T** = Create Excel Table (with dynamic ranges) from a Proper Data Set.
    i. Keyboard to name Excel Table: **Alt, J, T, A**
    ii. **Tab** = Enter Raw Data into an Excel Table.
24) **Ctrl + Shift + ~ (`)** = General Number Formatting Keyboard.
25) **Ctrl + ;** = Keyboard for hardcoding today's date.
26) **Ctrl + Shift + ;** = Keyboard for hardcoding current time.
27) **Arrow Key** = If you are making a formula, Arrow key will “hunt” for Cell Reference.
28) **Ctrl + B** = Bold the Font
29) **Ctrl + * (on Number Pad)** or **Ctrl + Shift + 8** = Highlight Current Table.
30) **Alt + Enter** = Add Manual Line Break (Word Wrap)
31) **Ctrl + P =** Print dialog Backstage View and Print Preview
32) **F4 Key** = If you are in Edit mode while making a formula AND your cursor is touching a particular Cell Reference, F4 key will toggle through the different Cell References:
    i. **A1** = Relative
    ii. **$A$1** = Absolute or “Locked”

   iii. **A$1** = Mixed with Row Locked (Relative as you copy across the columns AND Locked as you copy down the rows)
   iv. **$A1** = Mixed with Column Locked (Relative as you copy down the rows AND Locked as you across the columns)
33) **Ctrl + Shift + 4** = Apply Currency Number Formatting
34) **Tab key** = When you are selecting a Function from the Function Drop-down list, you can select the function that is highlighted in blue by using the Tab key.
35) **F9 Key** = To evaluate just a single part of formula while you are in edit mode, highlight part of formula and hit the F9 key.
   i. If you are creating an Array Constant in your formula: Hit F9.
   ii. If you are evaluating the formula element just to see what that part of the formula looks like, REMEMBER: to Undo with Ctrl + Z.
36) **Alt, E, A, A** = Clear All (Content and Formatting)
37) Evaluate Formula One Step at a Time Keyboard: **Alt, M, V**
38) Keyboard to open Sort dialog box: **Alt, D, S**
39) **Ctrl + Shift + L** = Filter (or **Alt, D, F, F**) = Toggle key for Filter Drop-down Arrows
40) **Ctrl + N** = Open New File
41) **F12** = Save As (Change File Name, Location, File Type)
42) Import Excel Table into Power Query Editor: **Alt, A, P, T**
43) **Ctrl + 1 (When Chart element in selected):** Open Task Pane for Chart Element
44) **F4 Key** = If you are in Edit mode while making a formula AND your cursor is touching a particular Cell Reference, F4 key will toggle through the different Cell References:
   i. **A1** = Relative
   ii. **$A$1** = Absolute or "Locked"
   iii. **A$1** = Mixed with Row Locked (Relative as you copy across the columns AND Locked as you copy down the rows)
   iv. **$A1** = Mixed with Column Locked (Relative as you copy down the rows AND Locked as you across the columns)
45) Keyboard to open Scenario Manager = **Alt, T, E**
46) **Ctrl + Tab** = Toggle between Excel Workbook File Windows
47) **Ctrl + Shift + F3** = Create Names From Selection
48) **Ctrl + F3** = open Name Manager
49) **F3** = Paste Name or List of Names
50) **Alt + F4** = Close Active Window
51) **Window Key + Up Arrow** = Maximize Active Window
52) **Ctrl + Shift + Enter** = Keystroke to enter Array Formulas that: 1) have a function argument that requires it, or 2) whether or not you are entering the Resultant Array into multiple cells simultaneously.
53) **Ctrl + /** = Highlight current Array

New In This Video:

54) None